

# Introduction à Maple : le modèle IS-LM

Bernard Dupont

[Bernard.Dupont@univ-lille1.fr](mailto:Bernard.Dupont@univ-lille1.fr)

Ce chapitre introductif a pour but de donner un aperçu des possibilités offertes par Maple pour la modélisation macroéconomique. Il ne s'agit pas de faire le tour de toutes les capacités mathématiques de ce logiciel de calcul formel, mais de montrer ce que peut être un dossier de travail (*worksheet*) Maple en macroéconomie. Ce chapitre va porter sur trois aspects de la modélisation :

- 1- résolution d'un système d'équations
- 2- techniques analytiques usuelles sur les fonctions de plusieurs variables
- 3- représentation graphique.

Le modèle keynésien de court terme IS-LM (Investment Saving - Liquidity Money) sert de support à cette exploration dans la mesure où il est relativement simple, bien connu et que par conséquent il est possible de vérifier à tout moment la conformité des résultats fournis par Maple aux connaissances acquises. Mais il va donner une image incomplète des vastes possibilités offertes par Maple en économie puisqu'on verra dans d'autres chapitres que ce logiciel permet aussi de résoudre des problèmes d'optimisation et de dynamique.

## Forme structurelle du modèle IS-LM

La **forme structurelle** du modèle IS-LM basique comprend les 10 équations suivantes, dont on rappelle la signification économique :

1. Equilibre (égalité de l'offre globale et de la demande globale) sur le marché des biens et services :  $Y = C + Inv + G$
2. Fonction de consommation keynésienne :  $C = c(Y - T) + C_c$
3. Fonction d'investissement :  $Inv = Inv_c - \alpha i$
4. Exogénéité des dépenses publiques :  $G = G_c$
5. Fonction de recettes fiscales :  $T = t Y$
6. Equilibre sur le marché de la monnaie :  $M_o = M_d$
7. Exogénéité de l'offre de monnaie :  $M_o = M_c$
8. Définition de la demande de monnaie par les agents économiques :  $M_d = L_1 + L_2$
9. Demande de monnaie de transaction :  $L_1 = l_1 Y$
10. Demande de monnaie de spéculation :  $L_2 = L_2 c - l_2 i$

Ce système de 10 équations a 10 inconnues, appelées en économie les **variables endogènes**, à savoir :

1. Le niveau d'activité (PIB en économie fermée) :  $Y$
2. La consommation des ménages :  $C$
3. L'investissement des entreprises :  $Inv$
4. Les dépenses publiques :  $G$
5. Les prélèvements fiscaux :  $T$
6. L'offre de monnaie :  $M_o$
7. La demande de monnaie :  $M_d$
8. La demande de monnaie de transaction :  $L_1$
9. La demande de monnaie de spéculation :  $L_2$
10. Le taux d'intérêt :  $i$

Interviennent également 5 **variables exogènes** que le modèle n'est pas chargé d'expliquer et considère comme prédéterminées à un niveau strictement positif :

1. La consommation autonome :  $C_c$
2. L'investissement autonome :  $Inv_c$
3. Les dépenses publiques autonomes :  $G_c$
4. L'offre de monnaie exogène :  $M_c$
5. La demande de monnaie autonome :  $L2_c$

Enfin, on note la présence de 5 **paramètres** sur lesquels la théorie économique (et le bon sens) pose des hypothèses :

1. La propension moyenne à consommer, comprise entre 0 et 1 :  $c$
2. La sensibilité de l'investissement au taux d'intérêt, strictement positive :  $\alpha$
3. Le taux moyen d'imposition, compris entre 0 et 1 :  $t$
4. La sensibilité de la demande de monnaie de transaction au niveau d'activité, strictement positive :  $l1$
5. La sensibilité de la demande de monnaie spéculative au taux d'intérêt, strictement positive :  $l2$

En Maple, on écrit directement le système d'équations annotées comme suit :

```
> eq1:=Y=C+Inv+G; #équilibre sur le marché des produits
eq2:=C=c*(Y-T)+Cc;#fonction de consommation keynésienne
eq3:=Inv=Inv_c-alpha*i;#fonction d'investissement
eq4:=G=G_c;#exogénéité des dépenses publiques
eq5:=T=t*Y;#recettes fiscales
eq6:=Mo=Md;#équilibre sur le marché de la monnaie
eq7:=Mo=M_c;#exogénéité de l'offre de monnaie
eq8:=Md=L1+L2;#définition de la demande de monnaie
eq9:=L1=l1*Y;#motif de transaction
eq10:=L2=L2_c-l2*i;#motif de spéculation
```

```
eq1 := Y = C + Inv + G
eq2 := C = c (Y - T) + Cc
eq3 := Inv = Inv_c - alpha i
eq4 := G = G_c
eq5 := T = t Y
eq6 := Mo = Md
eq7 := Mo = M_c
eq8 := Md = L1 + L2
eq9 := L1 = l1 Y
eq10 := L2 = L2_c - l2 i
```

Chaque équation porte un nom, de **eq1** à **eq10**. L'assignation se fait par la combinaison de touches "deux points+égal", soit **:=**, qu'il ne faut en aucun cas réduire à **=**, symbole informatique de l'égalité de deux membres. Du coup, chaque équation est bien écrite comme égalité de deux membres à l'aide du signe égal. Toute instruction élémentaire est close par un point virgule **;**. Les annotations servent à documenter les lignes et n'ont aucun intérêt mathématique. C'est pourquoi elles sont précédées du symbole **#** qui signale au logiciel qu'il ne doit pas en tenir compte. Pour passer à la ligne suivante, on utilise la combinaison de touches "Maj+Return".

On regroupe les divers ingrédients du modèle par catégories : variables endogènes, variables

exogènes, paramètres. Une catégorie est assignée à un nom de code évocateur et ses éléments sont regroupés dans un ensemble marqué par des accolades  $\{ \}$  et séparés par des virgules :

```
> endo := {Y, C, Inv, G, T, Mo, Md, L1, L2, i}; #ensemble des 10
endogènes
exo := {Cc, Invc, Gc, Mc, L2c}; #ensemble des 5 exogènes
para := {c, alpha, t, l1, l2}; #ensemble des paramètres
      endo := {C, G, Inv, L1, L2, Md, Mo, T, Y, i}
      exo := {Cc, Gc, L2c, Mc, Invc}
      para := {c, l1, l2, t, alpha} (1.1)
```

Il reste à lister les restrictions économiques sur les valeurs que peuvent prendre les paramètres et les exogènes. La commande **assume** sert à poser une hypothèse de travail. Elle ne renvoie pas de résultat.

```
> assume(0 < c, c < 1); #propension marginale à consommer
assume(Cc > 0); #consommation autonome
assume(Invc > 0); #investissement privé et public autonome
assume(alpha > 0); #sensibilité de l'investissement au niveau
de l'activité
assume(Gc > 0); #dépense publique autonome
assume(0 < t, t < 1); #taux de l'imposition compris entre 0 et 1
assume(Mc > 0); #exogénéité de l'offre de monnaie
assume(l1 > 0); #sensibilité de la demande de monnaie au
niveau d'activité
assume(l2 > 0); #sensibilité de la demande de monnaie au taux
d'intérêt
assume(L2c > 0); #demande de monnaie autonome
```

## Forme réduite du modèle IS-LM

La forme structurelle du modèle IS-LM est un système de 10 équations à 10 inconnues. Donner sa forme réduite consiste à exprimer les 10 endogènes en fonction des exogènes et des paramètres. Même si le système est linéaire, il faut avouer que c'est un travail particulièrement fastidieux où les risques d'erreurs de calcul sont très importants (essayez!). Avec Maple, on passe de la forme structurelle à la forme réduite à l'aide d'une seule commande, à savoir **solve** :

```
> soluce := solve({eq1, eq2, eq3, eq4, eq5, eq6, eq7, eq8, eq9, eq10},
endo);
```

$$soluce := \left\{ C = \frac{1}{l_2 - c l_2 + c t l_2 + \alpha l_1} (c \sim Invc \sim l_2 + c \sim \alpha \sim Mc \sim - c \sim \alpha \sim L2c \sim + c \sim Gc \sim l_2 - c \sim t \sim Invc \sim l_2 - c \sim t \sim \alpha \sim Mc \sim + c \sim t \sim \alpha \sim L2c \sim - c \sim t \sim Gc \sim l_2 + Cc \sim l_2 + Cc \sim \alpha \sim l_1), G = Gc, Inv = - \frac{1}{l_2 - c l_2 + c t l_2 + \alpha l_1} (- Invc \sim l_2 + c \sim Invc \sim l_2 - c \sim t \sim Invc \sim l_2$$

$$\begin{aligned}
& -\alpha \sim Mc \sim + c \sim \alpha \sim Mc \sim - c \sim t \sim \alpha \sim Mc \sim + Cc \sim \alpha \sim l1 \sim + \alpha \sim l1 \sim Gc \sim + \alpha \sim L2c \sim \\
& - c \sim \alpha \sim L2c \sim + c \sim t \sim \alpha \sim L2c \sim), L1 \\
& = \frac{l1 \sim (Cc \sim l2 \sim + Invc \sim l2 \sim + \alpha \sim Mc \sim - \alpha \sim L2c \sim + Gc \sim l2 \sim)}{l2 \sim - c \sim l2 \sim + c \sim t \sim l2 \sim + \alpha \sim l1 \sim}, L2 = \\
& - \frac{1}{l2 \sim - c \sim l2 \sim + c \sim t \sim l2 \sim + \alpha \sim l1 \sim} (-Mc \sim l2 \sim + Mc \sim c \sim l2 \sim - Mc \sim c \sim t \sim l2 \sim \\
& + l1 \sim Cc \sim l2 \sim + l1 \sim Invc \sim l2 \sim - l1 \sim \alpha \sim L2c \sim + l1 \sim Gc \sim l2 \sim), Md = Mc \sim, Mo \\
& = Mc \sim, T = \frac{t \sim (Cc \sim l2 \sim + Invc \sim l2 \sim + \alpha \sim Mc \sim - \alpha \sim L2c \sim + Gc \sim l2 \sim)}{l2 \sim - c \sim l2 \sim + c \sim t \sim l2 \sim + \alpha \sim l1 \sim}, Y \\
& = \frac{Cc \sim l2 \sim + Invc \sim l2 \sim + \alpha \sim Mc \sim - \alpha \sim L2c \sim + Gc \sim l2 \sim}{l2 \sim - c \sim l2 \sim + c \sim t \sim l2 \sim + \alpha \sim l1 \sim}, i \\
& = \frac{1}{l2 \sim - c \sim l2 \sim + c \sim t \sim l2 \sim + \alpha \sim l1 \sim} (-Mc \sim + Mc \sim c \sim - Mc \sim c \sim t \sim + l1 \sim Cc \sim \\
& + l1 \sim Invc \sim + l1 \sim Gc \sim + L2c \sim - L2c \sim c \sim + L2c \sim c \sim t \sim) \}
\end{aligned}$$

Maple résout quasi instantanément le système! Les valeurs des endogènes équilibrant simultanément les marchés des biens et services, de la monnaie et des titres (par la loi de Walras) sont déterminées sans ambiguïté. On note la présence d'un tilde ~ juste après les paramètres et exogènes ayant fait l'objet d'hypothèses.

Bien entendu, on s'intéresse essentiellement au niveau d'activité  $Y$  qu'on récupère et assigne comme suit.

```
> soluceY:=subs(soluce,Y);
```

$$soluceY := \frac{Cc \sim l2 \sim + Invc \sim l2 \sim + \alpha \sim Mc \sim - \alpha \sim L2c \sim + Gc \sim l2 \sim}{l2 \sim - c \sim l2 \sim + c \sim t \sim l2 \sim + \alpha \sim l1 \sim}$$

Quelques manipulations vont lui donner son apparence usuelle. Les premières lignes d'instructions se terminent par deux point (:) pour empêcher l'affichage des outputs. La dernière ligne réassigne la variable **soluceY** et demande l'affichage du résultat par un point virgule (;).

```
> numerY:=numer(soluceY):#récupération du numérateur
numer0Y:=sort(expand(numerY/l2),[Cc,Invc,Gc,L2c,Mc])
:#manipulations sur le numérateur
denomY:=denom(soluceY):#récupération du dénominateur
denom0Y:=collect(denomY/l2,[c,t,l1],distributed)
:#manipulations sur le dénominateur
soluceY:=numer0Y/denom0Y;
```

$$soluceY := \frac{Cc \sim + Invc \sim + Gc \sim - \frac{\alpha \sim L2c \sim}{l2 \sim} + \frac{\alpha \sim Mc \sim}{l2 \sim}}{-c \sim + c \sim t \sim + 1 + \frac{\alpha \sim l1 \sim}{l2 \sim}} \quad (2.1)$$

Le dénominateur joue un rôle crucial dans la théorie keynésienne. Il est à la base des effets multiplicateurs et partant de l'efficacité des politiques économiques. On s'assure tout d'abord par

un test que le dénominateur est strictement positif. La commande **is** est très instable et donne des résultats qui peuvent différer d'une session à l'autre ou d'un ordinateur à l'autre. Il est bon de recouper les informations qu'elle renvoie comme c'est le cas ci-dessous sachant que la réponse n'est jamais fausse.

```
> is(denom0Y>0);#test sur le signe du dénominateur
is(1-c+c*t>0);#test sur la première partie du dénominateur
is(alpha*l1/l2>0);#test sur la deuxième partie du
dénominateur
FAIL
true
true (2.2)
```

Maple est basé sur une logique ternaire. Une proposition est soit vraie (**true**), fausse (**false**) ou indécidable(**FAIL**). A la première question "le denominateur est-il strictement positif?", il arrive que la réponse soit certaines fois "FAIL/indécidable" et à d'autres "true/vrai". C'est décevant et frustrant. Tout rentre dans l'ordre en testant séparément **1-c+c\*t** et **alpha\*l1/l2** : le dénominateur est la somme de deux quantités strictement positives; il est donc strictement positif, tout comme son inverse, qui représente le multiplicateur "basique" qui sera interprété plus loin comme le multiplicateur budgétaire.

Testons à présent la position du dénominateur par rapport à la valeur 1.

```
> is(denom0Y<1);
FAIL (2.3)
```

La proposition est indécidable. Sans hypothèses supplémentaires sur les sensibilités **α**, **l1** et **l2**, on ne peut pas être assuré que le dénominateur est inférieur à 1, et donc que le multiplicateur basique est supérieur à 1.

Afin de préparer l'analyse d'impact, on exprime maintenant le niveau d'activité d'équilibre comme une **fonction** des exogènes et des paramètres :

```
> fY:=unapply(soluceY,Cc,Invc,Gc,Mc,t,c,alpha,l1,l2,L2c);#le
produit d'équilibre est une fonction de 10 variables
```

$fY := (Cc\sim, Invc\sim, Gc\sim, Mc\sim, t\sim, c\sim, \alpha\sim, l1\sim, l2\sim, L2c\sim)$

$$\rightarrow \frac{Cc\sim + Invc\sim + Gc\sim - \frac{\alpha\sim L2c\sim}{l2\sim} + \frac{\alpha\sim Mc\sim}{l2\sim}}{-c\sim + c\sim t\sim + 1 + \frac{\alpha\sim l1\sim}{l2\sim}}$$

On est alors en mesure de calculer tous les multiplicateurs.

L'impact des "vraies" exogènes tout d'abord est mesuré par le multiplicateur proprement dit. Les deux commandes suivantes montrent qu'il est égal au multiplicateur basique. Comme on a montré que ce dernier est strictement positif, l'accroissement des vraies exogènes a un effet positif sur le revenu d'équilibre.

```
> Diff(fY(Cc,Invc,Gc,Mc,t,c,alpha,l1,l2,L2c),Cc):%=value(%);
#Multiplicateur de la consommation autonome
```

$$\frac{\partial}{\partial Cc\sim} \left( \frac{Cc\sim + Invc\sim + Gc\sim - \frac{\alpha\sim L2c\sim}{l2\sim} + \frac{\alpha\sim Mc\sim}{l2\sim}}{-c\sim + c\sim t\sim + 1 + \frac{\alpha\sim l1\sim}{l2\sim}} \right) = \frac{1}{-c\sim + c\sim t\sim + 1 + \frac{\alpha\sim l1\sim}{l2\sim}}$$

```
> Diff(fY(Cc,Invc,Gc,Mc,t,c,alpha,l1,l2,L2c),Invc):%=value(%);
#Multiplicateur d'investissement
```

$$\frac{\partial}{\partial \text{Invc}^{\sim}} \left( \frac{C^{\sim} + \text{Invc}^{\sim} + G^{\sim} - \frac{\alpha^{\sim} L2c^{\sim}}{l2^{\sim}} + \frac{\alpha^{\sim} Mc^{\sim}}{l2^{\sim}}}{-c^{\sim} + c^{\sim} t^{\sim} + 1 + \frac{\alpha^{\sim} l1^{\sim}}{l2^{\sim}}} \right)$$

$$= \frac{1}{-c^{\sim} + c^{\sim} t^{\sim} + 1 + \frac{\alpha^{\sim} l1^{\sim}}{l2^{\sim}}}$$

Passons aux "fausses" exogènes que sont les mesures de politique économique (politique budgétaire, politique monétaire et politique fiscale).

> **Diff(fY(Cc,Invc,Gc,Mc,t,c,alpha,l1,l2,L2c),Gc):%=value(%) ;**  
**#Multiplicateur budgétaire**

$$\frac{\partial}{\partial G^{\sim}} \left( \frac{C^{\sim} + \text{Invc}^{\sim} + G^{\sim} - \frac{\alpha^{\sim} L2c^{\sim}}{l2^{\sim}} + \frac{\alpha^{\sim} Mc^{\sim}}{l2^{\sim}}}{-c^{\sim} + c^{\sim} t^{\sim} + 1 + \frac{\alpha^{\sim} l1^{\sim}}{l2^{\sim}}} \right) = \frac{1}{-c^{\sim} + c^{\sim} t^{\sim} + 1 + \frac{\alpha^{\sim} l1^{\sim}}{l2^{\sim}}}$$

Le multiplicateur budgétaire, égal au multiplicateur basique, est positif comme on l'a vu plus haut.

> **Diff(fY(Cc,Invc,Gc,Mc,t,c,alpha,l1,l2,L2c),t):%=value(%) ;**  
**#Multiplicateur fiscal**  
**is(denom(rhs(%))>=0);#test (superflu!) sur le signe du**  
**dénominateur du multiplicateur fiscal**  
**is(-c\*(Cc+Invc+Gc-alpha\*L2c/l2+alpha\*Mc/l2)<0) assuming -**  
**alpha\*L2c/l2+alpha\*Mc/l2>0;#test sous condition**  
**supplémentaire sur le signe du numérateur fiscal**

$$\frac{\partial}{\partial t^{\sim}} \left( \frac{C^{\sim} + \text{Invc}^{\sim} + G^{\sim} - \frac{\alpha^{\sim} L2c^{\sim}}{l2^{\sim}} + \frac{\alpha^{\sim} Mc^{\sim}}{l2^{\sim}}}{-c^{\sim} + c^{\sim} t^{\sim} + 1 + \frac{\alpha^{\sim} l1^{\sim}}{l2^{\sim}}} \right) =$$

$$- \frac{\left( C^{\sim} + \text{Invc}^{\sim} + G^{\sim} - \frac{\alpha^{\sim} L2c^{\sim}}{l2^{\sim}} + \frac{\alpha^{\sim} Mc^{\sim}}{l2^{\sim}} \right) c^{\sim}}{\left( -c^{\sim} + c^{\sim} t^{\sim} + 1 + \frac{\alpha^{\sim} l1^{\sim}}{l2^{\sim}} \right)^2}$$

true  
true

Fort logiquement, le multiplicateur fiscal est négatif (à condition que la demande autonome de monnaie soit inférieure à l'offre de monnaie).

> **Diff(fY(Cc,Invc,Gc,Mc,t,c,alpha,l1,l2,L2c),Mc):%=value(%) ;**  
**#impact d'un accroissement de la masse monétaire**  
**is(alpha/l2>0);#test sur le signe du numérateur du**  
**multiplicateur de masse monétaire**

$$\frac{\partial}{\partial Mc^{\sim}} \left( \frac{C^{\sim} + \text{Invc}^{\sim} + G^{\sim} - \frac{\alpha^{\sim} L2c^{\sim}}{l2^{\sim}} + \frac{\alpha^{\sim} Mc^{\sim}}{l2^{\sim}}}{-c^{\sim} + c^{\sim} t^{\sim} + 1 + \frac{\alpha^{\sim} l1^{\sim}}{l2^{\sim}}} \right)$$

$$= \frac{\alpha}{l_2 \left( -c + c t + 1 + \frac{\alpha l_1}{l_2} \right)}$$

true

Un accroissement de la masse monétaire a un effet stimulant sur le niveau d'activité.

Pour le plaisir, mesurons des effets des variations de paramètres :

```
> Diff(fY(Cc,Invc,Gc,Mc,t,c,alpha,l1,l2,L2c),c):%=value(%);
#Modification de la propension à consommer
is(-(-1+t)*(Cc+Invc+Gc-alpha*L2c/l2+alpha*Mc/l2)>0)
assuming -alpha*L2c/l2+alpha*Mc/l2>0;#test sous condition
supplémentaire sur le signe du numérateur
```

$$\frac{\partial}{\partial c} \left( \frac{Cc + Invc + Gc - \frac{\alpha L2c}{l_2} + \frac{\alpha Mc}{l_2}}{-c + c t + 1 + \frac{\alpha l_1}{l_2}} \right) =$$

$$\frac{\left( Cc + Invc + Gc - \frac{\alpha L2c}{l_2} + \frac{\alpha Mc}{l_2} \right) (-1 + t)}{\left( -c + c t + 1 + \frac{\alpha l_1}{l_2} \right)^2}$$

true

Résultat conforme aux préceptes keynésiens : une augmentation de la propension à consommer accroît le produit d'équilibre.

Les effets des autres paramètres sont bien moins simples :

```
> Diff(fY(Cc,Invc,Gc,Mc,t,c,alpha,l1,l2,L2c),alpha):%=value
(%);#Modification de la sensibilité de l'investissement au
niveau d'activité
```

$$\frac{\partial}{\partial \alpha} \left( \frac{Cc + Invc + Gc - \frac{\alpha L2c}{l_2} + \frac{\alpha Mc}{l_2}}{-c + c t + 1 + \frac{\alpha l_1}{l_2}} \right) = \frac{-\frac{L2c}{l_2} + \frac{Mc}{l_2}}{-c + c t + 1 + \frac{\alpha l_1}{l_2}}$$

$$- \frac{\left( Cc + Invc + Gc - \frac{\alpha L2c}{l_2} + \frac{\alpha Mc}{l_2} \right) l_1}{\left( -c + c t + 1 + \frac{\alpha l_1}{l_2} \right)^2 l_2}$$

```
> Diff(fY(Cc,Invc,Gc,Mc,t,c,alpha,l1,l2,L2c),l1):%=value(%);
#Modification de la sensibilité de la demande de monnaie au
niveau d'activité
```

$$\frac{\partial}{\partial l_1} \left( \frac{Cc + Invc + Gc - \frac{\alpha L2c}{l_2} + \frac{\alpha Mc}{l_2}}{-c + c t + 1 + \frac{\alpha l_1}{l_2}} \right) =$$

$$\frac{\left( Cc\sim + Invc\sim + Gc\sim - \frac{\alpha\sim L2c\sim}{l2\sim} + \frac{\alpha\sim Mc\sim}{l2\sim} \right) \alpha\sim}{\left( -c\sim + c\sim t\sim + 1 + \frac{\alpha\sim l1\sim}{l2\sim} \right)^2 l2\sim}$$

> Diff(fY(Cc,Invc,Gc,Mc,t,c,alpha,l1,l2,L2c),l2):%=value(%);  
**#Modification de la sensibilité la demande de monnaie au  
taux d'intérêt**

$$\frac{\partial}{\partial l2\sim} \left( \frac{Cc\sim + Invc\sim + Gc\sim - \frac{\alpha\sim L2c\sim}{l2\sim} + \frac{\alpha\sim Mc\sim}{l2\sim}}{-c\sim + c\sim t\sim + 1 + \frac{\alpha\sim l1\sim}{l2\sim}} \right) = \frac{\frac{\alpha\sim L2c\sim}{l2\sim^2} - \frac{\alpha\sim Mc\sim}{l2\sim^2}}{-c\sim + c\sim t\sim + 1 + \frac{\alpha\sim l1\sim}{l2\sim}}$$

$$+ \frac{\left( Cc\sim + Invc\sim + Gc\sim - \frac{\alpha\sim L2c\sim}{l2\sim} + \frac{\alpha\sim Mc\sim}{l2\sim} \right) \alpha\sim l1\sim}{\left( -c\sim + c\sim t\sim + 1 + \frac{\alpha\sim l1\sim}{l2\sim} \right)^2 l2\sim^2}$$

On pourrait obtenir aussi rapidement des informations sur les autres endogènes, en particulier le taux d'intérêt. L'intérêt de Maple est manifeste. En quelques lignes de programmation, on est parvenu à résoudre en un temps record un système de 10 équations à 10 inconnues, puis à mettre en évidence tous les effets multiplicateurs possibles et imaginables (y compris l'impact des paramètres structurels) sur le niveau d'activité. D'ailleurs, au prix d'un petit effort supplémentaire, on tire la matrice des 50 multiplicateurs d'impact mesurant l'impact des 5 exogènes sur les 10 endogènes.

> with(LinearAlgebra):#appel au paquetage LinearAlgebra  
**Soluce:=convert(soluce,list):**  
**Endo:=Vector([seq(lhs(Soluce[k]),k=1..10)]);#vecteur des**  
**endogènes**  
**Exo:=convert(convert(exo,list),Vector);#vecteur des**  
**exogènes**  
**Mul,End:=GenerateMatrix(Soluce,convert(exo,list))**  
**:#génération automatique de (l'opposé de) la matrice des**  
**multiplicateurs**  
**-Mul;#matrice des multiplicateurs à 10 lignes et 5 colonnes**

$$Endo := \begin{bmatrix} C \\ G \\ Inv \\ L1 \\ L2 \\ Md \\ Mo \\ T \\ Y \\ i \end{bmatrix}$$



$$Exo := \begin{bmatrix} Cc\sim \\ Gc\sim \\ Mc\sim \\ Invc\sim \\ L2c\sim \end{bmatrix}$$

$$\begin{aligned} & \left[ \left[ \frac{l2\sim + \alpha\sim l1\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, \frac{c\sim l2\sim - c\sim t\sim l2\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, \right. \right. \\ & \left. \frac{c\sim \alpha\sim - c\sim t\sim \alpha\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, \frac{c\sim l2\sim - c\sim t\sim l2\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, \right. \\ & \left. \frac{-c\sim \alpha\sim + c\sim t\sim \alpha\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim} \right], \\ & \left[ 0, 1, 0, 0, 0 \right], \\ & \left[ -\frac{\alpha\sim l1\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, -\frac{\alpha\sim l1\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, \right. \\ & -\frac{c\sim \alpha\sim - c\sim t\sim \alpha\sim - \alpha\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, -\frac{-l2\sim + c\sim l2\sim - c\sim t\sim l2\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, \\ & \left. -\frac{\alpha\sim - c\sim \alpha\sim + c\sim t\sim \alpha\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim} \right], \\ & \left[ \frac{l1\sim l2\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, \frac{l1\sim l2\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, \right. \\ & \frac{\alpha\sim l1\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, \frac{l1\sim l2\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, \\ & \left. -\frac{\alpha\sim l1\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim} \right], \\ & \left[ -\frac{l1\sim l2\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, -\frac{l1\sim l2\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, \right. \\ & -\frac{-l2\sim + c\sim l2\sim - c\sim t\sim l2\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, -\frac{l1\sim l2\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, \\ & \left. \frac{\alpha\sim l1\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim} \right], \\ & \left[ 0, 0, 1, 0, 0 \right], \\ & \left[ 0, 0, 1, 0, 0 \right], \\ & \left[ \frac{t\sim l2\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, \frac{t\sim l2\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, \right. \\ & \frac{t\sim \alpha\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, \frac{t\sim l2\sim}{l2\sim - c\sim l2\sim + c\sim t\sim l2\sim + \alpha\sim l1\sim}, \end{aligned} \tag{2.4}$$

$$\left. \begin{aligned} & - \frac{t \sim \alpha \sim}{l2 \sim - c \sim l2 \sim + c \sim t \sim l2 \sim + \alpha \sim l1 \sim} \right\} \\ & \left[ \frac{l2 \sim}{l2 \sim - c \sim l2 \sim + c \sim t \sim l2 \sim + \alpha \sim l1 \sim}, \frac{l2 \sim}{l2 \sim - c \sim l2 \sim + c \sim t \sim l2 \sim + \alpha \sim l1 \sim}, \right. \\ & \frac{\alpha \sim}{l2 \sim - c \sim l2 \sim + c \sim t \sim l2 \sim + \alpha \sim l1 \sim}, \frac{l2 \sim}{l2 \sim - c \sim l2 \sim + c \sim t \sim l2 \sim + \alpha \sim l1 \sim}, \\ & \left. - \frac{\alpha \sim}{l2 \sim - c \sim l2 \sim + c \sim t \sim l2 \sim + \alpha \sim l1 \sim} \right\} \\ & \left[ \frac{l1 \sim}{l2 \sim - c \sim l2 \sim + c \sim t \sim l2 \sim + \alpha \sim l1 \sim}, \frac{l1 \sim}{l2 \sim - c \sim l2 \sim + c \sim t \sim l2 \sim + \alpha \sim l1 \sim}, \right. \\ & \frac{-c \sim t \sim - 1 + c \sim}{l2 \sim - c \sim l2 \sim + c \sim t \sim l2 \sim + \alpha \sim l1 \sim}, \frac{l1 \sim}{l2 \sim - c \sim l2 \sim + c \sim t \sim l2 \sim + \alpha \sim l1 \sim}, \\ & \left. \frac{1 - c \sim + c \sim t \sim}{l2 \sim - c \sim l2 \sim + c \sim t \sim l2 \sim + \alpha \sim l1 \sim} \right] \end{aligned}$$

Bien entendu, les enseignements de macroéconomie ne vont jamais aussi loin dans la résolution du modèle IS-LM. Guidés par des considérations mélangeant la pédagogie et la théorie économique, ils centrent tout le travail d'analyse des propriétés sur les relations entre le revenu  $Y$  et le taux d'intérêt  $i$ .

## ▼ Forme semi-réduite du modèle IS-LM

On étudie maintenant la forme dite semi-réduite qui reprend la présentation usuelle du modèle IS-LM dans les manuels de macroéconomie. L'idée générale est distinguer deux blocs (IS et LM) en séparant la relation existant entre le niveau d'activité  $Y$  et le taux d'intérêt  $i$  sur le marché des produits d'une part et le marché de la monnaie d'autre part.

### ▼ Résolution du bloc IS en $Y$ et $i$

On extrait la relation entre  $Y$  et  $i$  à partir des 5 équations concernant le marché des biens et services :

```
> sol1:=solve({eq1,eq2,eq3,eq4,eq5},{Y,C,Inv,G,T});
```

$$sol1 := \left\{ \begin{aligned} & = \frac{c \sim Inv \sim - c \sim \alpha \sim i + c \sim G \sim - c \sim t \sim Inv \sim + c \sim t \sim \alpha \sim i - c \sim t \sim G \sim + C \sim}{1 - c \sim + c \sim t \sim}, G \\ & = G \sim, Inv = Inv \sim - \alpha \sim i, T = \frac{t \sim (C \sim + Inv \sim - \alpha \sim i + G \sim)}{1 - c \sim + c \sim t \sim}, Y \\ & = \frac{C \sim + Inv \sim - \alpha \sim i + G \sim}{1 - c \sim + c \sim t \sim} \end{aligned} \right\}$$

```
> sol1Y:=subs(sol1,Y):sol1Y:=collect(%,i);
```

$$sol1Y := - \frac{\alpha \sim i}{1 - c \sim + c \sim t \sim} + \frac{C \sim + Inv \sim + G \sim}{1 - c \sim + c \sim t \sim}$$

### ▼ Résolution du bloc LM en $Y$ et $i$

De la même manière, on extrait la relation entre  $Y$  et  $i$  à partir des équations concernant le

marché de la monnaie :

```
> sol2:=solve({eq6,eq7,eq8,eq9,eq10},{Mo,Md,L1,L2,i});
```

```
sol2 := { L1 = l1~ Y, L2 = Mc~ - l1~ Y, Md = Mc~, Mo = Mc~, i =  
-  $\frac{Mc~ - l1~ Y - L2c~}{l2~}$  }
```

```
> sol2Y:=i=subs(sol2,i):sol2Y:=collect(solve(sol2Y,Y),i);
```

```
sol2Y :=  $\frac{l2~ i}{l1~} + \frac{Mc~ - L2c~}{l1~}$ 
```

## ▼ Résolution graphique du modèle IS-LM

Afin de donner une représentation graphique de l'équilibre simultané sur les marchés des produits et de la monnaie, on attribue des valeurs aux exogènes et aux paramètres :

```
> val:={Cc=100,c=0.75,Invc=150,alpha=100,t=0.2,Gc=1000,Mc=500,  
l1=0.25,l2=200,L2c=120};
```

```
val := { Cc~ = 100, Gc~ = 1000, Mc~ = 500, c~ = 0.75, l1~ = 0.25, l2~ = 200, t~ = 0.2,  
Invc~ = 150, L2c~ = 120,  $\alpha$ ~ = 100 }
```

```
> IS:=subs(val,sol1Y);
```


```
IS := -250.0000000 i + 3125.000000
```

```
> LM:=subs(val,sol2Y);
```


```
LM := 800.0000000 i + 1520.000000
```

Sur le marché des produits, le niveau d'activité est une fonction affine décroissante du taux d'intérêt alors que sur le marché de la monnaie, la relation est affine croissante. Le tracé séparé des courbes IS et LM donne donc :

```
> grafIS:=plot(IS,i=0..3,color=red,thickness=3):grafIS;  
#construction de la courbe IS
```




```
> grafLM:=plot(LM,i=0..3,color=blue,thickness=3):grafLM;  
#conxtruction de la courbe LM
```



On visualise à présent l'équilibre à l'intersection des courbes IS et LM dans le plan  $(i,Y)$  :

```
> with(plots):display([grafIS,grafLM,textplot([1.5,2900,"E(0)"],font=[HELVETICA,OBLIQUE,10]),textplot([2.8,2550,"IS"],font=[HELVETICA,BOLDOBLIQUE,10]),textplot([2.45,3700,"LM"],font=[HELVETICA,BOLDOBLIQUE,10])]);
```



L'équilibre initial est noté  $E(0)$  dans ce graphique. On peut le calculer aisément afin de mesurer par la suite l'impact des mesures de politique économique sur son niveau.

Le taux d'intérêt et le revenu national d'équilibre sont solutions du système  $\{ Y = 3125 - 250 i, Y = 800 i + 1520 \}$  :

```
> solve({Y=IS,Y=LM},{Y,i});
equilibre_init:= [subs(% ,Y),subs(% ,i)];#calcul de
l'équilibre initial afin de préparer les exercices de
simulation
```

```
{Y=2742.857143, i=1.528571429}
```

```
equilibre_init := [2742.857143, 1.528571429]
```

## ▼ Effets de politique économique

Cette sous-section développe quelques scénarios de politique économique : accroissement budgétaire, accroissement de la fiscalité et accroissement de la masse monétaire.

### ▼ Politique budgétaire


On mesure et on visualise l'impact d'une augmentation des dépenses publiques, qui passent de 1000 à 1200.

```
> valG:={Cc=100,c=0.75,Inv=150,alpha=100,t=0.2,Gc=1200,
Mc=500,l1=0.25,l2=200,L2c=120};
```

```
valG := {Cc~ = 100, Gc~ = 1200, Mc~ = 500, c~ = 0.75, l1~ = 0.25, l2~ = 200, t~
```


```
=0.2, Invc~ = 150, L2c~ = 120,  $\alpha$ ~ = 100}
```

```
> ISG:=subs(valG,sol1Y);  
      ISG := -250.0000000 i + 3625.000000  
> grafISG:=plot(ISG,i=0..3,color=green,thickness=3)  
:grafISG;#graphe de la nouvelle courbe IS
```



```
> with(plottools):#appel du paquetage plottools pour  
inclure des flèches dans le graphique final  
f1:= arrow([0.5, 3100], [0.5, 3400],0.05,0.2,0.2,color=  
black):  
f2:= arrow([2.8, 2550], [2.8, 2850],0.05,0.2,0.2,color=  
black):  
> display([grafIS,grafISG,grafLM,f1,f2,textplot([1.5,  
2900,"E(0)"],font=[HELVETICA,OBLIQUE,10]),textplot([2,  
3300,"E(1)"],font=[HELVETICA,OBLIQUE,10]),textplot(  
[2.8,2300,"IS(0)"],font=[HELVETICA,BOLDOBLIQUE,10]),  
textplot([2.45,3700,"LM"],font=[HELVETICA,BOLDOBLIQUE,  
10]),textplot([2.8,3050,"IS(1)"],font=[HELVETICA,  
BOLDOBLIQUE,10])],title="Fig.1. Impact d'une hausse de  
G",titlefont=[HELVETICA,BOLDOBLIQUE,12]);
```

**Fig.1. Impact d'une hausse de G**



Le taux d'intérêt et le revenu national d'équilibre sont maintenant solutions du système  $\{ Y = 3625 - 250 i, Y = 800 i + 1520 \}$  :

```
> solve({Y=ISG,Y=LM},{Y,i});equilibre_1:=[subs(% ,Y),subs
(% ,i)];
      {Y= 3123.809524, i= 2.004761905}
equilibre_1 := [3123.809524, 2.004761905]
```

L'augmentation des dépenses publiques s'est traduite par une augmentation du revenu d'équilibre et du taux d'intérêt qu'on calcule aisément par différence entre l'état terminal et l'état initial :

```
> etat1:=equilibre_1-equilibre_init;
      etat1 := [380.952381, 0.476190476]
```


### ▼ Politique fiscale

On mesure et on visualise l'impact d'une augmentation du taux d'imposition, qui passe de 0.2 à 0.25.

```
> valt:={Cc=100,c=0.75,Invc=150,alpha=100,t=0.25,Gc=1000,
      Mc=500,l1=0.25,l2=200,L2c=120};
valt:= {Cc~ = 100, Gc~ = 1000, Mc~ = 500, c~ = 0.75, l1~ = 0.25, l2~ = 200, t~
      = 0.25, Invc~ = 150, L2c~ = 120, alpha~ = 100}
> Ist:=subs(valt,sol1Y);
      Ist := -228.5714286 i + 2857.142857
```




```
> grafIst:=plot(Ist,i=0..3,color=green,thickness=3)
:grafIst;
```



```
> f3:= arrow([0.5, 2950], [0.5, 2800],0.05,0.2,0.2,color=
black):
f4:= arrow([2.8, 2400], [2.8, 2250],0.05,0.2,0.2,color=
black):
```

```
> display([grafIS,grafIst,grafLM,f3,f4,textplot([1.5,
2900,"E(0)"],font=[HELVETICA,OBLIQUE,10]),textplot(
[1.3,2400,"E(2)"],font=[HELVETICA,OBLIQUE,10]),textplot(
[2.8,2600,"IS(0)"],font=[HELVETICA,BOLDOBLIQUE,10]),
textplot([2.8,2100,"IS(2)"],font=[HELVETICA,
BOLDOBLIQUE,10]),textplot([2.5,3700,"LM"],font=
[HELVETICA,BOLDOBLIQUE,10]),title="Fig.2. Impact d'une
hausse de t",titlefont=[HELVETICA,BOLDOBLIQUE,12]);
```

**Fig.2. Impact d'une hausse de  $t$**



Le taux d'intérêt et le revenu national d'équilibre sont maintenant solutions du système  $\{ Y = 2857, 14 - 258, 57 i, Y = 800 i + 1520 \}$  :

```
> solve({Y=ISt,Y=LM},{Y,i});equilibre_2:=[subs(% ,Y),subs
(% ,i)];
      {Y=2560.000000,i=1.300000000}
equilibre_2 := [2560.000000,1.300000000]
```

L'augmentation du taux d'imposition s'est traduite par une baisse du revenu d'équilibre et du taux d'intérêt :


```
> etat2:=equilibre_2-equilibre_init;
      etat2 := [-182.857143, -0.228571429]
```

### ▼ Politique monétaire

On mesure et on visualise l'impact d'une augmentation de la masse monétaire, qui passe de 500 à 600.


```
> valM:={Cc=100,c=0.75,Invc=150,alpha=100,t=0.25,Gc=1000,
      Mc=600,l1=0.25,l2=200,L2c=120};
valM := {Cc~ = 100, Gc~ = 1000, Mc~ = 600, c~ = 0.75, l1~ = 0.25, l2~ = 200, t~
      = 0.25, Invc~ = 150, L2c~ = 120, alpha~ = 100}
> LMM:=subs(valM,sol2Y);
      LMM := 800.0000000 i + 1920.000000
> grafLMM:=plot(LMM,i=0..3,color=green,thickness=3)
```

```
:grafLMM;
```



```
> f5:= arrow([0.5, 2000], [0.5, 2250],0.05,0.2,0.2,color=  
black):  
f6:= arrow([2.8, 3850], [2.8, 4100],0.05,0.2,0.2,color=  
black):  
> display([grafIS,grafLM,grafLMM,f5,f6,textplot([1.55,  
2600,"E(0)"],font=[HELVETICA,OBLIQUE,10]),textplot(  
[1.1,3020,"E(4)"],font=[HELVETICA,OBLIQUE,10]),textplot  
([2.8,2600,"IS"],font=[HELVETICA,BOLDOBLIQUE,10]),  
textplot([2.7,3450,"LM(0)"],font=[HELVETICA,  
BOLDOBLIQUE,10]),textplot([2.45,4100,"LM(4)"],font=  
[HELVETICA,BOLDOBLIQUE,10]),title="Fig.3. Impact d'une  
hausse de M",titlefont=[HELVETICA,BOLDOBLIQUE,12]);
```

**Fig.3. Impact d'une hausse de M**



Le taux d'intérêt et le revenu national d'équilibre sont maintenant solutions du système  $\{ Y = 3125 - 250i, Y = 800i + 1920 \}$  :

```
> solve({Y=IS,Y=LMM},{Y,i});equilibre_3:=[subs(% ,Y),subs
(% ,i)];
```

```
{Y=2838.095238, i=1.147619048}
```

```
equilibre_3 := [2838.095238, 1.147619048]
```

L'augmentation de la masse monétaire s'est traduite par une baisse du taux d'intérêt et une hausse du revenu d'équilibre :

```
> etat3:=equilibre_3-equilibre_init;
etat3 := [95.238095, -0.380952381]
```

## Conclusion

En quelques commandes, on est parvenu à mettre en évidence les principales propriétés formelles du modèle IS-LM, que ce soit en utilisant les ressources de l'algèbre linéaire ou en se laissant guider par la tradition de l'enseignement de macroéconomie. Maple est garant de leur exactitude (à condition que le programmeur soit à la hauteur des possibilités immenses du logiciel). En revanche, il ne peut en aucun cas interpréter ou même aider à interpréter les résultats obtenus. C'est là le véritable travail de l'économiste : faire des hypothèses pertinentes sur l'activité économique; proposer un modèle formel conforme à cet ensemble d'hypothèses; examiner de manière critique les résultats amenés par le "moulinage" du modèle. En clair, Maple "mouline" bien, mais il ne mouline que le grain qu'on lui a donné.